Calling qfloat floating-point functions (LCC-WIN32) from Maple

As an example the cumulative normal distribution cdfN is chosen (given through the error function).

The library qfloat.dll should be in Window's system directory and the concurrent cdfn_mpl.dll should be places in the directory of this worksheet.

The C source for the DLL is given at the end.

AVt, Dec 2005

```
> restart;
  kernelopts(version);
```

Maple 10.02, IBM INTEL NT, Nov 8 2005 Build ID 208934

```
> Digits_lcc:=115;
```

$$\text{Digits\_lcc} := 115$$

```
> Digits:=2*Digits_lcc; # greater precision to check results
```

$$\text{Digits} := 230$$

Define the cumulative normal distribution within Maple

```
> cdfN:= x -> 1/2+1/2*erf(1/2*x*2^(1/2));
```

$$\text{cdfN} := x \rightarrow \frac{1}{2} + \frac{1}{2}\,\text{erf}\left(\frac{1}{2}\,x\,\sqrt{2}\right)$$

For using the cdfn_mpl.dll locate its directory and call the external functions from there

```
> currentdir(): myDLL:=cat(%,`\\cdfn_mpl.dll`);
```

myDLL := "C:_Work\other\LCC_Work\cdfN_mpl\lcc\cdfn_mpl.dll"

A quick and dirty way is through strings:

```
> fct := define_external(
    'str_cdfN_str',
    'C',
    'x_str'::string[],
    'y_str'::string[],
    'nChar'::integer[4],
    RETURN::integer[4],
    LIB=myDLL):
```

This will provide the DLL with memory space (given as a string y_str) to store the results in the DLL. Since update is 'inplace' this will modify the string and its length. As Maple is a symbolic system one should never call this result directly, since this inconsistency for the same object will crash it (just try it and restart ...).

But a simple procedure solves the problem:

```
> cdfN_lcc:=proc(x)
    local X::string, Y;
    Y:=StringTools:-Fill( `0` , Digits_lcc);
    X:=convert(x,string);
    fct(X,Y,StringTools:-Length(Y));
    parse(Y);
  end proc:
```

Test that for inputs (first display yL from DLL, then Maple's result yM):

```
> xTst:= -1.2;
  yL:= cdfN_lcc(xTst):
  evalf(cdfN(xTst)): yM:=evalf(%,105):
  yL; yM;
  `absolute error`=yL-yM;
  `relative error`=evalf((yL-yM)/yM,105);
```

xTst := -1.2

0.11506967022170826802222020695663514867544703533750454155126330309006011524965021240106261\
29851471926091647

0.11506967022170826802222020695663514867544703533750454155126330309006011524965021240106261\
29851471926091647

absolute error = 0.

relative error = 0.

```
> xTst:= -10.2;
  yL:=cdfN_lcc(xTst):
  evalf(cdfN(xTst)): yM:=evalf(%,105):
  yL; yM;
  `absolute error`=yL-yM;
  `relative error`=evalf((yL-yM)/yM,105);
```

xTst := -10.2

0.99136251225599990522258020308993721069202729237947514523944342724108445351771215489440471\
47882269785663070 10^{-24}

0.99136251225599990522258020308993721069202729237947514523944342724108445351771215492129471\
08944570635892458 10^{-24}

absolute error = -0.26889961062300850229388 10^{-106}

relative error = -0.27124246408217063708830781566771528562073534099448650966784928493886408156\
1871403739977550023398139850407 10^{-82}

In the latter case one sees that qfloat exactness of 104 digits for the error function for those beyond the decimal point, while
Maple has a different notion for software floating-point numbers using
Float(SFloatMantissa(x), SFloatExponent(x)) = x .

```
> SFloatMantissa(yL),SFloatExponent(yL);
  SFloatMantissa(yM),SFloatExponent(yM);
```

99136251225599990522258020308993721069202729237947514523944342724108445351771215489440474\
7882269785663070, -129

99136251225599990522258020308993721069202729237947514523944342724108445351771215492129470\
8944570635892458, -129

One just has to accept this. So for small arguments the library returns 0:

```
> xTst:= -21.98;
  `lcc gives` = cdfN_lcc(xTst);
  `Maple says` = evalf(cdfN(xTst),200);
```

xTst := -21.98

lcc gives = 0.

Maple says = 0.22373098716097322294900434537276112287724538312713917614516088863670910388342\
52240158209015783 10^{-106}

A more sound way of calling is to use byte arrays instead of strings on which the DLL should work.
The code in the DLL is the same, just the calling changes

```
> fct2 := define_external(
    'str_cdfN_str',
    'C',
    'x_str'::string[],
    'y_str'::ARRAY('datatype' = 'integer'[ 1 ], 'order' = 'C_order'),
    'nChar'::integer[4],
    RETURN::integer[4],
    LIB=myDLL):
```

In this case the byte array is just made a string through the function call and the function returns the length.

For simple use here is a procedure:

```
> cdfN_lcc2:=proc(x)
    local X::string, Y, YB;
    X:=convert(x,string);
    Y:=StringTools:-Fill( "0" , Digits_lcc );
    YB:=StringTools:-ToByteArray(Y);
    fct2(X,YB,StringTools:-Length(Y));
    StringTools:-FromByteArray(YB);
    parse(%);
  end proc:
```

and it gives the same result:

```
> xTst:= -1.2;
  cdfN_lcc2(xTst);
  cdfN_lcc(xTst);
```

$$xTst := -1.2$$

$$0.11506967022170826802222020695663514867544703533750454155126330309006011524965021240106260\backslash$$
$$29851471926091647$$

$$0.11506967022170826802222020695663514867544703533750454155126330309006011524965021240106260\backslash$$
$$29851471926091647$$

A more direct but technical variant would be

```
> cdfN_lcc3:=proc(x)
    local X::string, YB, Filler, nChar;
    X:=convert(x,string);
    Filler:=48; # which is 0
    YB:=Array(1.. Digits_lcc + 10, 'datatype' = 'integer'[ 1 ],
      'order' = 'C_order',fill=Filler );
    nChar:= op(2,ArrayDims(YB));
    fct2(X,YB,nChar);
    StringTools:-FromByteArray(YB);
    parse(%);
  end proc:
```

which also gives the same result

```
> xTst:= -10.2;
  cdfN_lcc3(xTst);
  cdfN_lcc(xTst);
```

$$xTst := -10.2$$

$$0.99136251225599990522258020308993721069202729237947514523944342724108445351771215489440474782269785663070\ 10^{-24}$$

$$0.99136251225599990522258020308993721069202729237947514523944342724108445351771215489440474782269785663070\ 10^{-24}$$

I have not tried to use a Maple generated wrapper to access the DLL as the above already works ...

Finally look how one could extend the function for large arguments using asymtotics, which is coded in the following function

```
> fctplus := define_external(
    'str_cdfNplus_str',
    'C',
    'x_str'::string[],
    'y_str'::string[],
    'nChar'::integer[4],
    RETURN::integer[4],
    LIB=myDLL):

  cdfNplus_lcc:=proc(x)
    local X::string, Y;
    Y:=StringTools:-Fill( `0` , Digits_lcc + 10 );
    X:=convert(x,string);
    fctplus(X,Y,StringTools:-Length(Y));
    parse(Y);
  end proc:
```

Switch to high precision (say: 2000 digits)

```
> remDigits := Digits:
  Digits := 2000;
```

$$Digits := 2000$$

```
> xTst:= -30.2;
  yL:=cdfNplus_lcc(xTst):
  evalf(cdfN(xTst)): yM:=evalf(%,105):
  yL; yM; #yL-yM;
```

$$xTst := -30.2$$

$$0.11842886497168323042646690716339408922951664901803307910611698654580478863095565674107338913834722621161\ 10^{-199}$$

$$0.11842914786109124901419039341253996859495219358218882392438696087549715086245784233034294674686829290731 4\ 10^{-199}$$

```
> xTst:= -90.123456789012345678901234567890123456789012345678901234567890;
  yL:=cdfNplus_lcc(xTst):
  evalf(cdfN(xTst)): yM:=evalf(%,105):
  yL; yM;
```

$$xTst := -90.123456789012345678901234567890123456789012345678901234567890$$

$$0.84054928522861689157194478247483099666185599496426076279639731619986744896154371019991511807783465434189 1\ 10^{-1766}$$

$$0.84054931069229716783095748313921469284781969845373904750506340784442086555009974710183696645443633573386 2\ 10^{-1766}$$

The leading zeros are correct and then the next 5 - 6 digits and due to the rude solution I used.

```
> Digits:=remDigits:
>
```

Sources for cdfn_mpl.dll

Export the functions explicitely through a .def file

```
#include <windows.h>
#include <math.h>
#include <qfloat.h>

BOOL WINAPI __declspec(dllexport) LibMain(HINSTANCE hDLLInst, DWORD
fdwReason, LPVOID lpvReserved)
{
  switch (fdwReason)
  {
    case DLL_PROCESS_ATTACH:
      break;
    case DLL_PROCESS_DETACH:
      break;
    case DLL_THREAD_ATTACH:
      break;
    case DLL_THREAD_DETACH:
      break;
  }
  return TRUE;
}

extern __declspec(dllexport) long __stdcall
str_cdfN_str (char* x_str, char* y_str, int nChar)
{
 qfloat result;
 qfloat X;

 asctoq(x_str,&X);
 result = (erfq(X/sqrtq(2.0q)) + 1.0q)/2.0q;
 qtoasc(&result, y_str, min(strlen(y_str), 115));
 return strlen(y_str);
}


qfloat asymptotic_cdfN(qfloat X)
{
 qfloat c; // 1/sqrt(2*Pi)
 qfloat pdfN_X;
 qfloat result;

 if ( X == 0.0q )
  {return 0.0q;}
 c = "0.398942280401432677939946059934381868475858631164934657665925829670\
65792589930183850125233907306936 4303026";
 pdfN_X = expq( - X*X / 2.0q ) * c;
 result = - X / (1.0q + X*X) * pdfN_X;

 if ( 0.0q < X )
  {return result = 1.0q + result;}
 return result;
}


extern __declspec(dllexport) long __stdcall
```

```
str_cdfNplus_str (char* x_str, char* y_str, int nChar)
{
 qfloat result;
 qfloat X;

 asctoq(x_str,&X);
 if ( 21.97q < abs(X) )
  { result = asymptotic_cdfN(X);}
 else
  { result = (erfq(X/sqrtq(2.0q)) + 1.0q)/2.0q;}

 qtoasc(&result, y_str, min(strlen(y_str), 115));
 return strlen(y_str);
}
```